
Chapter 7: Slowly I Turn, …  
  

In which we learn how to go from here to there and back again.  
  

Internal Subroutines  
  

Assembly language offers several alternatives for organizing a program into blocks of code 
that can be reused. Well-designed programs depend on this principle. In this chapter we consider 
internal subroutines.  An internal subroutine consists of code that resides in the same control section 
as the code that is calling it. Later, we will tackle external subroutines which can reside in 
separately assembled modules and then linked together with the calling code or called dynamically. 
The techniques for calling internal and external subroutines differ considerably owing to differences 
in where the modules reside. 

  
 Conceptually, internal subroutines are much simpler to create than external ones. We need 

only two commands:  BAS and BR. BAS stands for “branch and save”, but might better be named 
“save and branch” because BAS first saves the address of the next instruction (the one following the 
BAS) – this will act as a return address – in the operand 1 register. After saving the return address, a 
branch is taken to the address specified by operand 2. After the target routine has executed, BR 
(Branch Unconditionally to Register) provides an unconditional branch to the return address stored 
in operand 1 – a register. Here is an example of how these two commands can be used to create a call 
and return. 

 
          BAS  R4,THERE    SAVE RETURN ADDRESS IN R4 
                           BRANCH TO “THERE” 
          ...  
          BAS  R4,THERE    SAVE RETURN ADDRESS IN R4 
          ...              BRANCH TO “THERE” 
 
THERE     EQU  *           TARGET ROUTINE 
          ...              ROUTINE BODY CODE HERE 
          BR   R4          RETURN TO CALLER 

 
The first BAS branches to the subroutine after saving the return address (the next 

instruction address) in register 4. The subroutine executes and control is returned at the first return 
address when BR is executed. This acts as an unconditional branch to the address in R4. The process 
repeats with the second BAS after it saves a different return address. 

 
Subroutines can call other subroutines using the same technique and even the same register, 

but they will need to follow a protocol: 
 

1. Create a fullword in which to store the return address (close to the routine is a good 
place), 

2. Store the register used to save the return address (let’s call it a linkage register) 
upon entry to the routine, and  

3. Restore the linkage register before exiting the routine.   
 

We will cover the required instructions (ST - Store Fullword and L - Load Fullword) in a 
later chapter, but here’s a simple example for the impatient or curious.  
  



 
          BAS  R4,THERE    SAVE RETURN ADDRESS IN R4 
                           BRANCH TO “THERE” 
          ... 
 
THERE     EQU  *           TARGET ROUTINE 
          ST   R4,SAVE4    STORE THE RETURN ADDRESS 
          ...              ROUTINE BODY CODE HERE 
          BAS  R4,YON      CALL ANOTHER ROUTINE 
          L    R4,SAVE4 
          BR   R4          RETURN TO CALLER 
SAVE4     DS   F 
 
Upon entering the routine, we store the return address in SAVE4. Upon returning from the 

YON routine, we restore the return address for THERE by loading SAVE4. If all routines in a calling 
sequence follow the protocol, a single linkage register will support the flow of control for the entire 
sequence. 
 

We cover only two instructions in this chapter:  BAS and BR. The second instruction, BR, is 
an extended mnemonic instruction for the native BCR (Branch on Condition Register) which itself is 
the register version of another native instruction BC (Branch on Condition) that we covered in the 
chapter 5.  

 
To reiterate, BC and BCR are native instructions and you can read about them in Principles 

of Operation. There are numerous extended mnemonic instructions related to these instructions 
which make them easier to use. BR, the instruction we are using in this chapter, is one of them. You 
can read about extended mnemonics in the HLASM Language Reference.  



  

 
 
 

 
  

OP 

Code 
R1X2 B2D2 D2D2 

  
BAS is a RX instruction which is used to support internal subroutines. When executed, the 

address of the instruction which follows the BAS, a return address, is stored in the operand 1 
register, and a branch is taken to the address specified by operand 2. BAS is used in combination 
with BCR (technically, BR, an extended mnemonic of BCR) to construct internal subroutines 
(routines that are contained in the same control section).  

 
Consider the instruction sequence below  

  
               BAS R8,SUB1  
                        BAS R8,SUB2  
                        MVC X,Y                         
...  
                   SUB1 EQU *   
                        ... (SUBROUTINE CODE GOES HERE)  
                        BR R8 BRANCH TO THE RETURN ADDRESS   
                   SUB2 EQU *  
                        ... (SUBROUTINE CODE GOES HERE)  
                        BR R8 BRANCH TO THE RETURN ADDRESS   
 

When the first BAS instruction is executed, the address of the next instruction (BAS 
R8,SUB2)is loaded into R8. After this return address is loaded, a branch occurs to the address 
denoted by SUB1. This begins execution of the code in the subroutine. At completion of the 
subroutine, an unconditional branch (BR R8) returns control to the address in R8. Execution 
resumes at the second BAS instruction. BR is an extended mnemonic of BCR, and branches 
unconditionally to the address in the operand 1 register. The second BAS causes the address of the 
next instruction (MVC X,Y) to be loaded into R8. A branch is taken to the subroutine denoted by 
SUB2. After execution of the subroutine, the unconditional branch (BR R8) at the end of the 
subroutine returns control at the MVC instruction.  

  
   BAS replaces an older instruction called “BAL” which stands for “Branch and Link”. Both 

of these instructions load the address of the next instruction into operand 1. The difference in their 
operation depends on the addressing mode that the machine is using:  

  
o In 24-bit mode: BAL loads bits 0 - 7 of operand 1 with linkage information ( 

instruction length code, condition code, program mask )  
o BAL loads bits 8 - 31 of operand 1 with the 24-bit return address  
o BAS loads bits 0 - 7 with eight 0’s  

BAS  R1,D2(X2,B2) RX Branch and Save 

               



o BAS loads bits 8 - 31 of operand 1 with a 24-bit return address  
o In 31-bit mode BAS and BAL load bit 0 with a 1 indicating 31-bit mode addressing  
o BAS and BAL load bits 1 - 31 with a 31-bit return address  

 
   The information that was provided by BAL in bits 0 - 7, can now be obtained using the IPM 

(insert Program Mask) instruction.  
  

  
  
Some Unrelated BAS’s  
 
           BAS  R4,HERE     LOAD NEXT ADDRESS IN R4, BRANCH TO “HERE”  
          BAS  R8,THERE    LOAD NEXT ADDRESS IN R8, BRANCH TO “THERE”  
          BAS  R10,YON     LOAD NEXT ADDRESS IN R10, BRANCH TO “YON” 
  

 Tips 
  

1. Use BAS instead of BAL to avoid non-zero bits being placed in the high-order byte of the 
stored address.  

 
2. When creating internal subroutines, consider saving the linkage register on entry to the 

subroutine and restoring it just before exiting:  
 

SUB1  EQU  *  
ST  R8,SAVE8  SAVE THE RETURN ADDRESS LOCALLY  
...  (SUBROUTINE CODE GOES HERE)  
L  R8,SAVE8  RESTORE THE RETURN ADDRESS  
BR  R8  
DS  0F  

SAVE8 DS  F  
 

By saving and restoring the linkage register, the subroutine is free to call other internal 
subroutines with the same linkage register.  

  
Trying It Out in VisibleZ:   

1. Load the program bas.obj from the \Codes directory and single step through each 
instruction until you are about to execute the BAS instruction.  

2. Which register will be used to hold the return address?  
3. What is the return address?  
4. What is the index register?  
5. Where is the target of the address?  
6. Cycle through the BAS instruction.  What address was stored in R4?  Why?  
7. Cycle instructions until you are at BCR.  What is the mask for this BCR?  What are the 

conditions under which the branch will occur?    
8. We have executed a “subroutine” and now we are returning.  Is the return target correct?  

Examples 



  

 
 
 
 
 
 
 
 
 
 
 
 
 
  
 

The Branch on Condition Register instruction examines the 2-bit condition code in the PSW 
and branches (or not) based on the value it finds.  Operand 1 is a self-defining term which represents 
a 4-bit mask (binary pattern) indicating the conditions under which the branch should occur.  
Operand 2 is a register containing the target address to which the branch will be made if the 
condition indicated in Operand 1 occurs. If the condition code is one of the values specified in the 
mask, the instruction address in the PSW is replaced with the target address in R2.  This causes the 
processor to fetch the instruction located at the target address as the next instruction in the 
fetch/execute cycle.  

 
   There are four possible values for the condition code: 

 
                              Condition Code                  Meaning 
 
                   00            Zero or Equal 
                   01            Low or Minus 
                   10            High or Plus 
                   11            Overflow  
                   

When constructing a mask for Operand 1, each bit (moving from the high-order bit to the low-
order bit) represents one of the four conditions in the following order:  Zero/Equal, Low/Minus, High/Plus, 
Overflow.  Consider the following instructions, 
 
                           LA   3,THERE 
                           BCR  8,3 
 
In the BCR, the first operand,”8”, is a decimal self-defining term and represents the binary mask 
B’1000’.  Since the first bit is a 1, the mask indicates that a branch should occur on a zero or equal 
condition.  Since the other bits are all 0, no branch will be taken on the other conditions.  The first 
operand could be designated as any equivalent self-defining term.  For example, the following 
instruction is equivalent to the BCR above. 
 
                           BCR   B’1000’,3 
 

Extended mnemonics were developed to replace the awkward construction of having to code 
a mask.  The extended mnemonics are easier to code and read.  A listing of the extended mnemonics 
follows below. 
  

Branch On Condition 
Register BCR        M1,R2            RR 

  Op 
Code 

 

M1R2 
 



 
 
BER    Branch Equal Register     BNER    Branch Not Equal Register 
BZR    Branch Zero Register      BNZR    Branch Not Zero Register 
BLR    Branch Low Register       BNLR    Branch Not Low Register 
BMR    Branch Minus Register     BNMR    Branch Not Minus Register 
BHR    Branch High Register      BNHR    Branch Not High Register 
BPR    Branch Positive Register  BNPR    Branch Not Positive Register 
NOPR   No Operation Register     BR      Unconditional Branch Register 
 
 

Using extended mnemonics, we could replace the previous Branch On Condition Register 
instruction with the following, 

 
                           BZR   3 
 
The “BZR” means “Branch on Condition Zero Register”.  When the assembler processes this BZR, it 
generates the mask as B’1000’.  The table below indicates the possible mask values and the 
equivalent extended mnemonics. Notice that not all masks have an equivalent extended mnemonic. 
 
 Eq/Low   Low/Min   High/Plus   Overflow   Decimal     Extended 
                                           Condition   Mnemonic 
   0          0          0          0          0          NOPR  
   0          0          0          1          1          BOR 
   0          0          1          0          2          BHR, BPR 
   0          0          1          1          3          No mnemonic 
   0          1          0          0          4          BLR, BMR 
   0          1          0          1          5          No mnemonic 
   0          1          1          0          6          No mnemonic 
   0          1          1          1          7          BNER, BNZR 
   1          0          0          0          8          BER, BZR 
   1          0          0          1          9          No mnemonic 
   1          0          1          0         10          No mnemonic 
   1          0          1          1         11          BNLR, BNMR 
   1          1          0          0         12          No mnemonic 
   1          1          0          1         13          BNHR, BNPR 
   1          1          1          0         14          No mnemonic 
   1          1          1          1         15          BR 
  



 

 
 
 
 
          Some Unrelated Branch on Condition Register Examples 
 
              LA   7,THERE   POINT REGISTER AT TARGET ADDRESS 
              LTR  R8,R8     SET THE CONDITION CODE 
              BPR  7         BRANCH IF CONDITION CODE IS POSITIVE   
              ...            OTHERWISE FALL THROUGH TO NEXT INSTRUCTION   
      THERE   EQU  * 
 
              LA   5,THERE   POINT REGISTER AT TARGET ADDRESS 
              CLC  X,Y       SET THE CONDITION CODE 
              BER  5         BRANCH IF X = Y   
              ...            OTHERWISE FALL THROUGH TO NEXT INSTRUCTION   
      THERE   EQU  * 
  
 
              LA   9,THERE   POINT REGISTER AT TARGET ADDRESS 
              CLC  X,Y       SET THE CONDITION CODE 
              BHR  9         BRANCH IF X > Y   
              ...            OTHERWISE FALL THROUGH TO NEXT INSTRUCTION   
      THERE   EQU  * 
 
              LA   R6,THERE  LOAD R6 WITH THE TARGET ADDRESS 
              BR   R6        BRACH TO THE ADDRESS CONTAINED IN R6 
 
              THIS LAST EXAMPLE IS FUNCTIONALLY EQUIVALENT TO THIS EXAMPLE 
              WHICH DOESN’T USE A REGISTER EXPLICITLY: 
 
              B    THERE 
 
 

Examples 


